

1. Modulation examples

1.1 Simple AM:

```
instr 1

indx = p7 ; modulation index
imod = p6 ; modulation frequency
ifreq = p5 ; carrier frequency
iamp = p4 ; amplitude
idur = p3 ; total sound duration
irise = .01 ; attack time
idec = .05 ; decay time
itable = 1 ; sinewave table

amod oscil iamp*indx, imod, itable ; modulator
acarr oscil iamp+amod, ifreq, itable ; carrier
asig linen acarr, irise, idur, idec ; envelope

out asig
endin
```

1.2 Ring Modulation with a sine wave oscillator:

```
instr 1

imod = p5 ; modulation frequency
ifreq = p4 ; carrier frequency
idur = p3 ; total sound duration
itable = 1 ; sinewave table

asig in ; signal input
acarr oscil asig, ifreq, itable ; carrier

out acarr
endin
```

1.3 General-purpose ring modulation:

```
instr 1

asig1 soundin "file1.wav"
asig2 soundin "file2.wav"
out asig1*asig2
endin
```

1.4 Vibrato

```
instr 1

iwidth = p7 ; vibrato width
irate = p6 ; vibrato rate
ifreq = p5 ; frequency
iamp = p4 ; amplitude
idur = p3 ; total sound duration
irise = .01 ; attack time
idec = .05 ; decay time
isine = 1 ; sinewave table
iwave = 2 ; other wave table

kamp linen iamp, irise, idur, idec ; amp envelope
kvib oscil iwidth, irate, isine ; modulator
asig oscil kamp, ifreq+kvib, itwave ; oscillator
out asig
endin
```

1.5 A snare-drum design

This design is discussed in detail on Dodge & Jerse's book, chapter 3. Here's a brief description of its three components:

- (1) The basic noisy snare sound is obtained by ring modulating a noise source which energy mostly up to 400Hz (the randi noise generator). The ring modulation will centre the spectrum of the noise around the carrier frequency, 500Hz.
- (2) An extra set of inharmonic components is created by a second oscillator containing a wave with the harmonics 10, 16, 22, 23 with the fundamental at $1/10^{\text{th}}$ of the drum 'fundamental frequency'.
- (3) Finally, a sine wave is added at the drum 'fundamental'.

The envelopes controlling the amplitudes of these three elements are decaying exponential curves.

(instrument)

```
instr 1

iamp = p4 ; overall amplitude
ifreq = p5 ; frequency (skin tuning)
isine = 1 ; sinewave
inhar = 2 ; complex wave with harmonics 10 16 22 23

kamp1 expon iamp/2, p3, .00012*iamp/2 ; using exponential curve generator
kamp2 expon iamp/6, p3, .00012*iamp/6 ; to generate the time functions
kamp3 expon iamp/2.5, p3, .004*iamp/2.5 ; (envelopes)
```

```

aran  randi  kamp1, 400                ; band-limited noise
anoi  oscil  aran, 500, isine          ; ring-modulated (500Hz sine carrier)
ainh  oscil  kamp2, ifreq/10, 2       ; inharmonic components
afun  oscil  kamp3, ifreq, isine      ; sinewave generator

      out   afun+ainh+anoi            ; mix all the signals

endin

```

(score)

```

f1 0 1024 10 1 ; sine

;          GEN pn pi pp pn pi  pp pn pi  pp pn pi  pp
f2 0 1024  9 10 1 0 16 1.5 0  22 2 0  23 1.5 0 ; wave with partials 10, 16, 22
; and 23
; pn - partial number
; pi - partial intensity
; pp - partial phase

;          dur  amp  fc
i1 0      .5   10000 280
i1 .5     .5   15000 280
i1 1      .5   10000 280
i1 1.5    .5   15000 280
e

```

2. Sampled-sound synthesis examples

2.1 Using loscil

```

instr 1

ifreq = p5 ; frequency
iamp = p4 ; amplitude
idur = p3 ; total sound duration
irise = .001 ; attack time
idec = .05 ; decay time
iwave = 1 ; wave table

k1  linen iamp, irise, idur, idec ; line envelope
a1  loscil k1, ifreq, iwave ; looping oscillator
    out a1
endin

```

(score)

```

f1 0 0 1 [aiff filename goes here] 0 0 0 ; AIFF file with sample info

i1 0 3 2000 110 ; notes
i1 3 2 8000 220
i1 5 1 10000 440

```

2.1 Using a standard oscillator

```
instr 1

ispeed = p5 ; lookup speed
iamp = p4 ; amplitude
idur = p3 ; total sound duration
irise = .001 ; attack time
idec = .05 ; decay time
iwave = 1 ; wave table
itlen = ftlen(1) ; table length

k1 linen iamp, irise, idur, idec
a1 oscili k1, (ispeed*sr)/itlen, iwave
  out a1
endin
```

(score)

```
f1 0 65536 1 [filename goes here] 0 0 1
```

```
i1 0 5 20000 1 ; normal speed
i1 5 5 20000 1.5 ; 5th above
i1 10 5 20000 -1 ; reverse
```